

```

; 10.01.2009 HR
; Assembler ASIDE 1.14
; c't 8504
; dart_sio_ct8504
; Testprogramm für die I/O-Karte in Z80-Mnemonics
;
BASE:      EQU    080H      ; 4 DIL-SCHALTER

CTC:      EQU    BASE+00H  ;
CTC0:     EQU    BASE+0    ; ZAEHLER 0
CTC1:     EQU    BASE+1    ; ZAEHLER 1
CTC2:     EQU    BASE+2    ; ZAEHLER 2
CTC3:     EQU    BASE+3    ; ZAEHLER 3

SIO:      EQU    BASE+4H   ;
SIOAD:    EQU    SIO+0    ; KANAL A DATEN
SIOAC:    EQU    SIO+1    ; KANAL A CONT.
SIOBD:    EQU    SIO+2    ; KANAL B DATEN
SIOBC:    EQU    SIO+3    ; KANAL B CONT.

PIO1:     EQU    BASE+08H  ;
PIO1A:    EQU    PIO1+0    ; PIO 1 A DATEN
PIO1B:    EQU    PIO1+1    ; PIO 1 B DATEN
PIO1AC:   EQU    PIO1+2    ; PIO 1 A CONT.
PIO1BC:   EQU    PIO1+3    ; PIO 1 B CONT.

PIO2:     EQU    BASE+0CH  ;
PIO2A:    EQU    PIO2+0    ; PIO 2 A DATEN
PIO2B:    EQU    PIO2+1    ; PIO 2 B DATEN
PIO2AC:   EQU    PIO2+2    ; PIO 2 A CONT.
PIO2BC:   EQU    PIO2+3    ; PIO 2 B CONT.

          ORG    100H      ; PROGRAMM FUER CP/M 80

START:
LD        HL,(0000)
INC      HL
INC      HL
INC      HL
LD       DE,CONST
LD       BC,00009
LDIR
JR       XXXX

CONST:
JP       0                ; KBD-STATUS: CHR A=FF
                          ;          NO CHR A=00

CONIN:
JP       0                ; A: CHR

CONOUT:
JP       0                ; CHR. ZU; SCHIRM C:=CHR.

XXXX:
CALL    INIT
EI

LOOP:
IN      A,(PIO2B)        ; 7 AUSGABE PIO 2 B
INC     A                ; ERZEUGE RECHTECKE
AND     07FH             ; IM BINAER-CODE
OUT     (PIO2B),A

CALL    SIOIN            ; CHAR. VON SIO ?
JR      Z,LOOP1          ; NEIN
LD      C,A              ; CHAR. VON SIO AUF SCHIRM
DI
CALL    CONOUT           ;
EI

LOOP1:
CALL    CONST            ; CHAR. AUF KEYBOARD ?
OR      A                ;
JR      Z,LOOP           ; NEIN
CALL    CONIN            ;
CP      '.'              ; ABBRUCH ?
JR      NZ,LOOP2        ; NEIN

DI
NOP
NOP
JP      0000

```

```

LOOP2:
    LD    C,A          ;
    CALL SIOOUT        ; CHAR. VON KEYBOARD ZUM SIO
    JR    LOOP

SIOOUT:
    IN    A,(SIOBC)    ; STATUS LESEN
    BIT   2,A          ; TX SENDEPUFFER LEER ?
    JR    Z,SIOOUT     ; NEIN
    LD    A,A          ; CHAR.
    OUT   (SIOBD),A    ; SENDEN
    RET

SIOIN:
    IN    A,(SIOBC)    ; STATUS LESEN
    AND   0000001B     ; RX CHAR. VORHANDEN ?
    RET   Z            ; KEIN CHAR.
    IN    A,(SIOBD)    ;
    AND   01111111B    ; BIT 7 = 0
    RET

;-----

INIT:
    LD    HL,CTCIT     ; CTC 0 INITTALISIEREN
    CALL INITX

    LD    HL,SIOIT     ; SIO INITTALISIEREN
    CALL INITX

    LD    HL,PIO1T     ; PIO1 INITTALISIEREN
    CALL INITX

    LD    HL,PIO2T     ; PIO2 INITTALISIEREN
    CALL INITX

    LD    A,HI INTV0   ; CPU INITTALISIEREN
    LD    I,A
    IM    2

    RETI                ; GGF PORT-BAUSTEINE AUS INTER. HOLEN

INITX:
    LD    A,(HL)       ; BYTE-ANZAHL
    OR    A
    RET   Z
    LD    B,A
    INC  HL
    LD    C,(HL)
    INC  HL
    OTIR
    JR    INITX

CTCIT:
    DEFB 03,CTC0       ; 3 BYTE ZUM CTC0
    DEFB LO INTV8      ; INT-VEKTOR
    DEFB 10100111B     ; INT-ENABLE
                        ; RESET
                        ; ZEITGEBER / 256
                        ; ZEIT-KONST. FOLGT
    DEFB 255           ; ZEIT-KONST. 255

    DEFB 02,CTC1       ; 2 BYTE ZUM CTC1
    DEFB 00100111B     ; INT-DISABLE
                        ; RESET
                        ; ZEITGEBER / 256
                        ; ZEIT-KONST. FOLGT
    DEFB 255           ; ZEIT-KONST. 255

    DEFB 0              ; ENDE CTC-TABELLE

SIOIT:
    DEFB 11,SIOAC      ; 11 BYTE ZUM SIO A
    DEFB 00011000B     ; RESET
    DEFB 1,00000000B   ; REG1: ALLE INT. DISABLE
    DEFB 2,00000000B   ; REG2: INT.VEC. LEER
    DEFB 3,11000100B   ; REG3: 8 BIT/CHAR RX ENABLE
    DEFB 4,11000100B   ; REG4: TRANSMITTER X64 CLOCK
                        ; 1 STOPBIT NO PARITY

```

```

DEFB          5,01101010B          ; REG5; RECIVER DTR=0 8 BIT/CHAR
                                           ; TX ENABLE RTS=1

DEFB          0                    ; ENDE SIO-TABELLE

PIO1T:
DEFB          05,PIO1AC             ; 5 BYTE ZUM PIO 1 A CONTROL
DEFB          00000011B            ; DIS-INT
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          11001111B            ; BETR.ART BIT EIN/AUSG.
DEFB          00000000B            ; ALLES AUSG.

DEFB          01,PIO1A             ; 1 BYTE ZUM PIO 1 B DATEN
DEFB          00000000B            ; DATEN "LOW"

DEFB          0                    ; ENDE PIO-1-TABELLE

PIO2T:
DEFB          05,PIO2AC             ; 5 BYTE ZUM PIO 2 CONTROL
DEFB          00000011B            ; DIS-INT
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          11001111B            ; BETR.ART BIT EIN/AUSG.
DEFB          00000000B            ; ALLES AUSG.

DEFB          01,PIO2A             ; 1 BYTE ZUM PIO 2 A DATEN
DEFB          00000000B            ; DATEN "LOW"

DEFB          09,PIO2AC             ; 7 BYTE ZUM PIO 2 B CONTROL
DEFB          00000011B            ; DIS-INT
DEFB          LO INTVF             ; INT.-VECTOR
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          01001111B            ; BETR.ART BYTE EINGABE
DEFB          11001111B            ; BETR.ART EIN/AUSG.
DEFB          10000000B            ; 1 EINGANG SONST ALLES AUSG.
DEFB          00110111B            ; DIS-INT / AND HIGH / INT-MASKE
DEFB          01111111B            ; BIT 7 MACHT INTERRUPT
DEFB          10000011B            ; INT-ENABLE

DEFB          01,PIO2B             ; 1 BYTE ZUM PIO 2 B DATEN
DEFB          00000000B            ; DATEN "LOW"

DEFB          0                    ; ENDE PIO-2-TABELLE

;=====
ORG           200H                 ; VECTOR-TAB. ADR.LOW: XXXX XXX0

INTV0:
DEFW          RETIN                ; SIO
INTV1:
DEFW          RETIN
INTV2:
DEFW          RETIN
INTV3:
DEFW          RETIN
INTV4:
DEFW          RETIN
INTV5:
DEFW          RETIN
INTV6:
DEFW          RETIN
INTV7:
DEFW          RETIN

INTV8:
DEFW          FLAG0                ; CTC
INTV9:
DEFW          FLAG1
INTVA:
DEFW          FLAG2
INTVB:
DEFW          FLAG3

INTVC:
DEFW          RETIN                ; PIO 1
INTVD:
DEFW          RETIN

```

```
INTVE:      DEFW          RETIN          ; PIO 2
INTVF:      DEFW          FLAG4
```

```
;/=====
```

```
FLAG0:      PUSH         AF
             PUSH         BC
             LD           C,'0'
             CALL        CONOUT
             POP          BC
             POP          AF
```

```
RETIN:      EI
             RETI
```

```
FLAG1:      PUSH         AF
             PUSH         BC
             LD           C,'1'
             CALL        CONOUT
             POP          BC
             POP          AF
             EI
             RETI
```

```
FLAG2:      PUSH         AF
             PUSH         BC
             LD           C,'2'
             CALL        CONOUT
             POP          BC
             POP          AF
             EI
             RETI
```

```
FLAG3:      PUSH         AF
             PUSH         BC
             LD           C,'3'
             CALL        CONOUT
             POP          BC
             POP          AF
             EI
             RETI
```

```
FLAG4:      PUSH         AF
             PUSH         BC
             LD           C,'4'
             CALL        CONOUT
             POP          BC
             POP          AF
             EI
             RETI
```

```
END
```